

Illustration 1 shows the pristine cube where each face has a different color. The six colors of the pristine cube are identified by their first letter: B, G, C, R, M, and Y for Blue, Green, Cyan, Red, Magenta, and Yellow, respectively.




Illustration 2: Cube after operation R+ for clockwise rotation of right face.

Illustration 2 shows the result after the right face has been rotated by 90 degrees clockwise. Notice that the right face proper is unchanged with its color "R" but that the edges of the right face as they appear on the sides of the front, top, back and bottom have all been shifted around, the "C's" from the front are now on the top, the "B's" of the top have been shifted to the back, etc.

The Program

Lines 210 and 220 defines the colors of the cube faces. Lines 1010 thru 1030 defines the array C(I,J,K) holding the color of each cubie in the cube.

The letters used to indicate the various cube facing operations are defined in the listing of the computer program, lines 1130 thru 1240.

Clockwise quarter turns are indicated by the plus sign following the face movement command, such as: F+, R+, L+, B+, U+, and D+ for the corresponding face. (U, up, designates the top and D, down, the bottom. Be careful not to use B for bottom.)

A minus is used to indicate the counterclockwise operation, such as F-, R-, L-, B-, U-, and D-.

As facing movements are made, the array colors are shifted in accordance with the data lines, lines 1510 thru 2050, provided for each movement, lines 3000 thru 3090.

The program is such that a combination of many operations can be entered at one time. See line 1110 and the operation symbol PP\$.




Illustration 3: Cube after the combined operation:

F+B-U+D-R+L-F+B-

Illustration 3 is the result of such a combined operation: namely, F+B-U+D-R+L-F+B-. (Note that in entering such data, each letter must be followed by either a plus or a minus.) The pattern in Illustration 3 is equivalent to what is called "Dots" by Hofstadter.

Many interesting mathematical features of the cube as discussed by Hofstadter can be investigated quickly using this program. For example, consider the operation pair R+U+. Such an operation naturally will mix up the faces and edges of the right and upper faces.

However, if this operation pair is repeated seven times, the edge cubies return to their initial positions and orientations.

To do the same with the corner cubies takes fifteen combined operations. It is apparent that if the operations are carried out seven times fifteen, or 105 times, the cube will return to its pristine form.

While doing these 105 combined operations by hand is obviously very tedious, it is easily accomplished on the computer. The program is such that after typing an operation such as R+U+, one need only press the {RETURN} key 105 times. An easier way is to type:

R+U+R+U+R+U+R+U+R+U+R+U+R+U+R+U+

and press {RETURN} fifteen times.

The Perspective Diagram

Since the flat cube is somewhat difficult to relate to its three-dimensional counterpart, the program also has a more realistic perspective diagram of the cube that can be displayed at any time.

This diagram, however, can show only the one orientation - with the front, upper and right faces visible. It should not be difficult to develop the other perspectives and making another menu selection to show the desired view. To save space and time, this exercise is left to the reader.

To develop the perspective drawing, we first draw the outline of the cube. Since ZBASIC has powerful line drawing routines, the original H-19 graphics characters that were used to draw the edges of the cube in the original program have been replaced with the much sharper looking ZBASIC LINE routines, lines 4010 thru 4050.

Next, lines 4060 thru 4340 place the colored cubies in place, using the ZBASIC LOCATE commands. While the PRINT command could be used to print an entire line at a time, using the LOCATE command and printing each individual cubie permits changing the cubie color as necessary and superimposing the cubies among the cube's outline.

Examples of such diagrams are shown in Illustrations 4, 5, and 6, which correspond to Illustrations 1, 2, and 3, respectively.

```

TOP      B
      B  B  B
      B  B  B
      C  B  R  R
      C C  R  R
      C C C R  R
      C C R  R
FRONT    C  R  RIGHT

```

Illustration 4: Perspective view corresponding to Illustration 1.

```

TOP      B  B
      B  B  C
      B  C
      C  C  R  R
      C C Y R  R
      C Y R  R
FRONT    Y  R  RIGHT

```

Illustration 5: View corresponding to Illustration 2.

```

TOP      C  C
      C  C  C
      C  C
      R  C  B
      R R  B B
      R C R B R B
      R R B B
FRONT    R  B  RIGHT

```

Illustration 6: View corresponding to Illustration 3.

Control Operations

Four control operations are defined:

0: When an operation is requested (line 1110) entering a "0" will return the cube to the initial condition, like "solving" the puzzle instantly.

1: Entering a "1" has the effect of providing a more realistic perspective diagram of the cube after any arbitrary set of operations.

2: Entering a "2" causes the instructions to be reviewed without disrupting the current configuration of the cube.

Q: Finally, entering a `Q' will permit you to Quit the program and return you to ZBASIC.

Lower case letters are also acceptable for input.

The program written in the article was written in Microsoft BASIC as implemented for the HDOS operating system on a Heathkit H8 computer with 40K memory, an H19 terminal and H17 disk drive.

The program listed here is for a color or monochrome H/Z-100 using ZBASIC.

A similar cube simulation program has been published by J.W. Pehouseek in Dr. Dobb's Journal.

References:

- Wall Street Journal*, March 5, 1981
- Reader's Digest*, May 1981
- Scientific American*, March 1981
- Time*, March 23, 1981
- Dr. Dobb's Journal*, July 1981

If you have any questions or comments, please email me at:

z100lifeline@swvagts.com

Cheers,

Steven W. Vagts




RUBIK.BAS - Rubik's Cube Simulation Program

```
10 ' CUBE PROGRAM - Simulates Rubik's Cube
20 ' Original program by A.A. Frost, Sedona, AZ, for MBASIC
30 ' and published in Spring 1982 issue of "Sextant" magazine.
40 ' Converted to ZBASIC for the Z-100 and enhanced by:
50 ' Steven W. Vagts
60 ' 211 Sean Way, Hendersonville, NC 28792

100 CLEAR: DEFINT A-Z
110 DIM C(6,3,3)
120 T$=" ": REM 10 Spaces

200 ' Define Colors
210 CL(1)=1: CL(2)=2: CL(3)=3
220 CL(4)=4: CL(5)=5: CL(6)=6
230 GOSUB 6000

1000 ' Print Diagram.
1010 FOR I=1 TO 6: FOR J=1 TO 3: FOR K=1 TO 3
1020 C(I,J,K)=CL(I)
1030 NEXT K: NEXT J: NEXT I
1040 GOTO 2200

1100 PRINT: PRINT "ENTER OPERATION (Press {2} to repeat the instructions)"
1110 INPUT " ";A$: IF A$="" GOTO 1120 ELSE PP$=A$: LL=LEN(A$)
1120 FOR II=1 TO LL STEP 2: OP$=MID$(PP$,II,2)
1130 IF OP$="F+" OR OP$="f+" GOTO 1500
1140 IF OP$="F-" OR OP$="f-" GOTO 1550
1150 IF OP$="R+" OR OP$="r+" GOTO 1600
1160 IF OP$="R-" OR OP$="r-" GOTO 1650
1170 IF OP$="U+" OR OP$="u+" GOTO 1700
1180 IF OP$="U-" OR OP$="u-" GOTO 1750
1190 IF OP$="L+" OR OP$="l+" GOTO 1800
1200 IF OP$="L-" OR OP$="l-" GOTO 1850
1210 IF OP$="D+" OR OP$="d+" GOTO 1900
1220 IF OP$="D-" OR OP$="d-" GOTO 1950
1230 IF OP$="B+" OR OP$="b+" GOTO 2000
1240 IF OP$="B-" OR OP$="b-" GOTO 2050
1250 IF OP$="0" GOTO 1000
1260 IF OP$="1" GOTO 4000
1270 IF OP$="2" THEN GOSUB 6000: GOTO 1040
1280 IF OP$="Q" OR OP$="q" GOTO 2500
1290 GOTO 1100

1500 RESTORE 1500: GOSUB 3000: GOTO 2400
1510 ' Data for F and F'
1520 DATA 3,1,3,3,1,1,3,3,1,3,3,3, 3,1,2,3,2,1,3,3,2,3,2,3
1530 DATA 1,3,3,2,1,3,6,1,1,4,3,1, 1,3,1,2,3,3,6,1,3,4,1,1
1540 DATA 1,3,2,2,2,3,6,1,2,4,2,1
1550 RESTORE 1500: GOSUB 3050: GOTO 2400

1600 RESTORE 1600: GOSUB 3000: GOTO 2400
1610 ' Data for R and R'
1620 DATA 4,1,3,4,1,1,4,3,1,4,3,3, 4,1,2,4,2,1,4,3,2,4,2,3
1630 DATA 1,2,3,3,2,3,6,2,3,5,2,1, 1,1,3,3,1,3,6,1,3,5,3,1
1640 DATA 1,3,3,3,3,3,6,3,3,5,1,1
1650 RESTORE 1600: GOSUB 3050: GOTO 2400

1700 RESTORE 1700: GOSUB 3000: GOTO 2400
1710 ' Data for U and U'
1720 DATA 1,1,3,1,1,1,1,3,1,1,3,3, 1,1,2,1,2,1,1,3,2,1,2,3
1730 DATA 3,1,3,4,1,3,5,1,3,2,1,3, 3,1,1,4,1,1,5,1,1,2,1,1
1740 DATA 3,1,2,4,1,2,5,1,2,2,1,2
1750 RESTORE 1700: GOSUB 3050: GOTO 2400
```

```

1800 RESTORE 1800: GOSUB 3000: GOTO 2400
1810 '   Data for L and L'
1820 DATA 2,1,3,2,1,1,2,3,1,2,3,3, 2,1,2,2,2,1,2,3,2,2,2,3
1830 DATA 1,1,1,5,3,3,6,1,1,3,1,1, 1,3,1,5,1,3,6,3,1,3,3,1
1840 DATA 1,2,1,5,2,3,6,2,1,3,2,1
1850 RESTORE 1800: GOSUB 3050: GOTO 2400

1900 RESTORE 1900: GOSUB 3000: GOTO 2400
1910 '   Data for D and D'
1920 DATA 6,1,3,6,1,1,6,3,1,6,3,3, 6,1,2,6,2,1,6,3,2,6,2,3
1930 DATA 2,3,1,5,3,1,4,3,1,3,3,1, 2,3,3,5,3,3,4,3,3,3,3,3
1940 DATA 2,3,2,5,3,2,4,3,2,3,3,2
1950 RESTORE 1900: GOSUB 3050: GOTO 2400

2000 RESTORE 2000: GOSUB 3000: GOTO 2400
2010 '   Data for B and B'
2020 DATA 5,1,3,5,1,1,5,3,1,5,3,3, 5,1,2,5,2,1,5,3,2,5,2,3
2030 DATA 1,1,1,4,1,3,6,3,3,2,3,1, 1,1,3,4,3,3,6,3,1,2,1,1
2040 DATA 1,1,2,4,2,3,6,3,2,2,2,1
2050 RESTORE 2000: GOSUB 3050: GOTO 2400

2200 '   "COLOR" PRINTING ROUTINE
2210 CLS: I=1: FOR J=1 TO 3: PRINT TAB(25);: FOR K=1 TO 3: GOSUB 5000
2220 PRINT SPC(1);: NEXT K: PRINT
2230 IF J<>3 THEN PRINT: NEXT J
2240 PRINT TAB(29)"UP" TAB(45)"CUBE COLORS"
2250 FOR J=1 TO 3: PRINT: PRINT SPC(9);
2260 FOR I=2 TO 5: FOR K=1 TO 3
2270 GOSUB 5000
2280 PRINT SPC(1);: NEXT K
2290 PRINT SPC(3);: NEXT I: PRINT: NEXT J
2300 PRINT SPC(12);"LEFT          FRONT          RIGHT          BACK"
2310 FOR J=1 TO 3: PRINT: PRINT TAB(25);
2320 I=6: FOR K=1 TO 3: GOSUB 5000
2330 PRINT SPC(1);: NEXT K: PRINT
2340 NEXT J: PRINT TAB(28)"DOWN"
2350 IF II>=LL THEN 1100

2400 NEXT II
2410 GOTO 2200

2500 END

3000 '   Clockwise Rotation
3010 FOR I=1 TO 5: READ I1,J1,K1,I2,J2,K2,I3,J3,K3,I4,J4,K4
3020 CT=C(I1,J1,K1): C(I1,J1,K1)=C(I2,J2,K2)
3030 C(I2,J2,K2)=C(I3,J3,K3): C(I3,J3,K3)=C(I4,J4,K4)
3040 C(I4,J4,K4)=CT: NEXT I: RETURN

3050 '   CounterClockwise Rotation
3060 FOR I=1 TO 5: READ I1,J1,K1,I2,J2,K2,I3,J3,K3,I4,J4,K4
3070 CT=C(I1,J1,K1): C(I1,J1,K1)=C(I4,J4,K4)
3080 C(I4,J4,K4)=C(I3,J3,K3): C(I3,J3,K3)=C(I2,J2,K2)
3090 C(I2,J2,K2)=CT: NEXT I: RETURN

4000 '   Routine for Perspective Diagram
4010 CLS: S=20: LINE (120,50)-(260,20),7: LINE (260,20)-(400,50)
4020 LINE (120,50)-(260,80): LINE (260,80)-(400,50)
4030 LINE (120,50)-(120,120): LINE (120,120)-(260,150)
4040 LINE (260,150)-(400,120): LINE (400,120)-(400,50)
4050 LINE (260,80)-(260,150): LOCATE 4,20: PRINT "TOP"
4060 LOCATE 4,33,0: I=1: J=1: K=1: GOSUB 4500
4070 LOCATE 5,28: J=2: GOSUB 4500
4080 LOCATE 5,38: J=1: K=2: GOSUB 4500
4090 LOCATE 6,23: J=3: K=1: GOSUB 4500
4100 LOCATE 6,33: J=2: K=2: GOSUB 4500

```

```

4110 LOCATE 6,43: J=1: K=3: GOSUB 4500
4120 LOCATE 7,28: J=3: K=2: GOSUB 4500
4130 LOCATE 7,38: J=2: K=3: GOSUB 4500
4140 LOCATE 8,33: J=3: K=3: GOSUB 4500
4150 LOCATE 9,19: I=3: J=1: K=1: GOSUB 4500
4160 LOCATE 9,47: I=4: K=3: GOSUB 4500
4170 LOCATE 10,24: I=3: K=2: GOSUB 4500
4180 LOCATE 10,42: I=4: GOSUB 4500
4190 LOCATE 11,19: I=3: J=2: K=1: GOSUB 4500
4200 LOCATE 11,29: J=1: K=3: GOSUB 4500
4210 LOCATE 11,37: I=4: K=1: GOSUB 4500
4220 LOCATE 11,47: J=2: K=3: GOSUB 4500
4230 LOCATE 12,24: I=3: K=2: GOSUB 4500
4240 LOCATE 12,42: I=4: GOSUB 4500
4250 LOCATE 13,19: I=3: J=3: K=1: GOSUB 4500
4260 LOCATE 13,29: J=2: K=3: GOSUB 4500
4270 LOCATE 13,37: I=4: K=1: GOSUB 4500
4280 LOCATE 13,47: J=3: K=3: GOSUB 4500
4290 LOCATE 14,24: I=3: K=2: GOSUB 4500
4300 LOCATE 14,42: I=4: GOSUB 4500
4310 LOCATE 15,29: I=3: K=3: GOSUB 4500
4320 LOCATE 15,37: I=4: K=1: GOSUB 4500
4330 LOCATE 17,18: PRINT "FRONT": LOCATE 17,44: PRINT "RIGHT"
4340 LOCATE 18,1,1: COLOR 7,0: PRINT
4350 PRINT "ENTER OPERATION (Press {RETURN} to return to the puzzle)";
4360 INPUT PP$: LL=LEN(PP$): GOTO 1120

4500 ON C(I,J,K) GOTO 4510,4520,4530,4540,4550,4560
4510 COLOR 1,0: PRINT "B";: COLOR 7,0: RETURN
4520 COLOR 2,0: PRINT "G";: COLOR 7,0: RETURN
4530 COLOR 3,0: PRINT "C";: COLOR 7,0: RETURN
4540 COLOR 4,0: PRINT "R";: COLOR 7,0: RETURN
4550 COLOR 5,0: PRINT "M";: COLOR 7,0: RETURN
4560 COLOR 6,0: PRINT "Y";: COLOR 7,0: RETURN

5000 ON C(I,J,K) GOTO 5010,5020,5030,5040,5050,5060
5010 COLOR 0,1: PRINT " B ";: COLOR 7,0: RETURN
5020 COLOR 0,2: PRINT " G ";: COLOR 7,0: RETURN
5030 COLOR 0,3: PRINT " C ";: COLOR 7,0: RETURN
5040 COLOR 0,4: PRINT " R ";: COLOR 7,0: RETURN
5050 COLOR 0,5: PRINT " M ";: COLOR 7,0: RETURN
5060 COLOR 0,6: PRINT " Y ";: COLOR 7,0: RETURN

6000 CLS: PRINT TAB(25)"RUBIK'S CUBE": PRINT
6010 PRINT T$"This program simulates the rotations of the faces of the cube."
6020 PRINT T$"It begins with the pristine or unscrambled cube, then as the"
6030 PRINT T$"faces are rotated, various patterns are produced. Then you can"
6040 PRINT T$"try solving the scrambled puzzle to the original solid colors."
6050 PRINT T$"The cube faces have the colors represented by the letters:"
6060 PRINT: PRINT T$" `B' = Blue      `G' = Green      `C' = Cyan"
6070 PRINT: PRINT T$" `R' = Red      `M' = Magenta    `Y' = Yellow"
6080 PRINT: PRINT TAB(40)"To continue, press {RETURN}." : Z$=INPUT$(1)

6100 CLS: PRINT "      CUBE FACE OPERATIONS": PRINT
6110 PRINT T$"The operations corresponding to the various possible cube face"
6120 PRINT T$"rotations are defined as follows:"
6130 PRINT: PRINT T$"      F+ and F-      rotate the front face 90 degrees"
6140 PRINT: PRINT T$"      B+ and B-      rotate the back face"
6150 PRINT: PRINT T$"      R+ and R-      rotate the right face"
6160 PRINT: PRINT T$"      L+ and L-      rotate the left face"
6170 PRINT: PRINT T$"      U+ and U-      rotate the upper face"
6180 PRINT: PRINT T$"      D+ and D-      rotate the down, or bottom face"
6190 PRINT: PRINT T$"The +/- rotate the face clockwise or counterclockwise."
6200 PRINT: PRINT: PRINT TAB(40)"To continue, press {RETURN}." : Z$=INPUT$(1)
6210 CLS: PRINT: PRINT "      ENTERING OPERATIONS": PRINT

```

```

6220 PRINT T$"When `OPERATION' is requested by the computer, just type in any"
6230 PRINT T$"of the above operations; F+, for example, and press {RETURN}."
6240 PRINT T$"Also, if you wish to carry out two or more operations before"
6250 PRINT T$"seeing the result, just type in the successive operations on"
6260 PRINT T$"the same line; R+L-, for example, will rotate right and left"
6270 PRINT T$"faces together. A whole line of operations can be input at once,"
6280 PRINT T$"if desired, but the computing time will be considerable.": PRINT
6290 PRINT T$"Once an operation, or a series of operations, has been entered,"
6300 PRINT T$"it can be repeated just by pressing {RETURN}, since the last"
6310 PRINT T$"entry is still stored in memory.": PRINT
6320 PRINT TAB(40)"To continue, press {RETURN}.": Z$=INPUT$(1)

6350 CLS: PRINT: PRINT " SPECIAL CONTROL OPERATIONS": PRINT
6360 PRINT: PRINT T$"When `OPERATION' is requested, enter:"
6370 PRINT: PRINT T$" `0' to produce the pristine cube."
6380 PRINT: PRINT T$" `1' to produce a perspective diagram for the current"
6390 PRINT: PRINT T$" state of the cube."
6400 PRINT: PRINT T$" `2' to restart the program at the instructions."
6410 PRINT: PRINT T$" `Q' or `q' to QUIT the program."
6420 PRINT: PRINT
6430 PRINT " NOW, HERE WE GO! You will now see a diagram for the cube in"
6440 PRINT " a folded-out view.
6450 PRINT TAB(40)"Just press {RETURN} to begin."
6460 Z$=INPUT$(1): REM Waits for a key to be pressed.
6470 RETURN

```

